

```

2830 GOTO 4590
2840 PRINT "Congradulations, Captain! The last klingon battle cruiser"
2850 PRINT "menacing the federation has been destroyed.":PRINT
2860 PRINT "Your efficiency rating is:";1000*(K7/(T-T0))^2:GOTO 2790
2870 ! SHORT RANGE SENSOR SCAN & STARTUP SUBROUTINE
2880 FOR I=S1-1 TO S1+1:FOR J=S2-1 TO S2+1
2890 IF INT(I+.5)<1 OR INT(I+.5)>8 OR INT(J+.5)<1 OR INT(J+.5)>8 THEN 2910
2900 A$=" B ":Z1=I:Z2=J:GOSUB 4050:IF Z3=1 THEN 2920
2910 NEXT J:NEXT I:D0=0:GOTO 2940
2920 D0=1:C$="DOCKED":E=30000:P=P0
2930 PRINT "Shields dropped for docking purposes":S=0:GOTO 2960
2940 IF K3>0 THEN C$="*RED*":GOTO 2960
2950 C$="GREEN":IF E<(E0*.1) THEN C$="YELLOW"
2960 IF D(2)>=0 THEN 2980
2970 PRINT:PRINT "*** Short Range Sensors are out ***":PRINT:RETURN
2980 O1$="-----":PRINT O1$:FOR I=1 TO 8
2990 FOR J=(I-1)*24+1 TO (I-1)*24+22 STEP 3:PRINT " ";
3000 IF MID$(Q$,J,3)=" " THEN PRINT " "; ELSE PRINT MID$(Q$,J,3);
3010 NEXT J
3020 ON I GOTO 3030,3040,3050,3060,3070,3080,3090,3100
3030 PRINT " Stardate ";INT(T*10)*.1:GOTO 3120
3040 PRINT " Condition ";C$:GOTO 3120
3050 PRINT " Quadrant ";Q1;" ";Q2:GOTO 3120
3060 PRINT " Sector ";S1;" ";S2:GOTO 3120
3070 PRINT " Photon Torpedoes ";INT(P):GOTO 3120
3080 PRINT " Total Energy ";INT(E+S):GOTO 3120
3090 PRINT " Shields ";INT(S):GOTO 3120
3100 PRINT " Klingons Remaining";INT(K9)
3120 NEXT I:PRINT O1$:RETURN
3130 REM LIBRARY COMPUTER CODE

3140 IF D(8)<0 THEN PRINT "Computer Disabled."
:PRINT "(Spock beat it at Mastermind...)":GOTO 730
3150 INPUT "Computer: Working. What is your command";A
3170 IF A<0 THEN 730
3180 IF A>5 THEN A=6
3190 PRINT:H8=1:ON A+1 GOTO 3300,3430,3520,3820,3580,3280,3200
3200 PRINT "Functions available from library computer:"
3210 PRINT " 0 - Cumulative galactic record"
3220 PRINT " 1 - Status report"
3230 PRINT " 2 - Photon torpedo data"
3240 PRINT " 3 - Starbase navigation data"
3250 PRINT " 4 - Direction/Distance calculator"
3260 PRINT " 5 - Galaxy 'Region Name' map":PRINT:GOTO 3150
3270 ! SETUP TO CHANGE CUM GAL RECORD TO GALAXY MAP
3280 H8=0:G5=1:PRINT " The Galaxy":GOTO 3330
3290 ! CUM GALACTIC RECORD
3300 PRINT:PRINT " ";
3310 PRINT "Computer Record of Galaxy for quadrant";Q1;" ";Q2:PRINT
3330 PRINT " 1 2 3 4 5 6 7 8"
3340 O1$="-----"
3350 PRINT O1$:FOR I=1 TO 8:PRINT I,:IF H8=0 THEN 3390
3360 FOR J=1 TO 8:PRINT " ";:IF Z(I,J)=0 THEN PRINT "***":GOTO 3380
3370 PRINT MID$(STR$(Z(I,J)+1000),3,3);
3380 NEXT J:GOTO 3410
3390 Z4=I:Z5=1:GOSUB 4100:J0=INT(15-.5*LEN(G2%)):PRINT TAB(J0);G2%;
3400 Z5=5:GOSUB 4100:J0=INT(39-.5*LEN(G2%)):PRINT TAB(J0);G2%;
3410 PRINT:PRINT O1$:NEXT I:PRINT:GOTO 730
3420 ! STATUS REPORT
3430 PRINT " Status Report ":X$="":IF K9>1 THEN X$="s"
3440 PRINT "Klingon";X$;" left: ";K9
3450 PRINT "Mission must be completed in";.1*INT((T0+T9-T)*10);"stardates"
3460 X$="s":IF B9<2 THEN X$="":IF B9<1 THEN 3490

3470 PRINT "The Federation is maintaining";B9;"starbase";X$;" in the galaxy"
3480 GOTO 2450
3490 PRINT "Your stupidity has left you on your own!"

```